```1c
    Таб = Новый ТабличныйДокумент;

    ИмяФайлаОтчета = КаталогВременныхФайлов() + "Report_123.tmp";
    Отчет = Справочники.ВнешниеОбработки.НайтиПоНаименованию("Остатки склада+Стоимость");

    ДвоичныеДанные = Отчет.ХранилищеВнешнейОбработки.Получить();
    ДвоичныеДанные.Записать(ИмяФайлаОтчета);

    НужныйОтчет = ВнешниеОтчеты.Создать(ИмяФайлаОтчета, ложь);
    СхемаКомпоновкиДанных = НужныйОтчет.ПолучитьМакет("ОсновнаяСхемаКомпоновкиДанных");
    Настройки = СхемаКомпоновкиДанных.НастройкиПоУмолчанию;
    НачалоПериода = Настройки.ПараметрыДанных.Элементы.Найти("НачалоПериода");
    НачалоПериода.Значение = НачалоДня(ТекущаяДата())-86400;;
    КонецПериода = Настройки.ПараметрыДанных.Элементы.Найти("КонецПериода");
    КонецПериода.Значение = КонецДня(ТекущаяДата())-80000;
    Склад = Настройки.ПараметрыДанных.Элементы.Найти("Склад");
    Склад.Значение = Справочники.Склады.НайтиПоКоду("000000343");

    ТипЦен = Настройки.ПараметрыДанных.Элементы.Найти("ТипЦен");
    ТипЦен.Значение = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("000000354");

    //Помещаем в переменную данные о расшифровке данных
    Расшифровка = Новый ДанныеРасшифровкиКомпоновкиДанных;
    //Формируем макет, с помощью компоновщика макета
    КомпоновщикМакета = Новый КомпоновщикМакетаКомпоновкиДанных;
    //Передаем в макет компоновки схему, настройки и данные расшифровки
    МакетКомпоновки = КомпоновщикМакета.Выполнить(СхемаКомпоновкиДанных, Настройки, Расшифровка);
    ВнешниеПараметры = Новый Структура;
    //Выполним компоновку с помощью процессора компоновки
    ПроцессорКомпоновкиДанных = Новый ПроцессорКомпоновкиДанных;
    ПроцессорКомпоновкиДанных.Инициализировать(МакетКомпоновки, ВнешниеПараметры, Расшифровка);
    //Выводим результат в табличный документ
    ПроцессорВывода = Новый ПроцессорВыводаРезультатаКомпоновкиДанныхВТабличныйДокумент;
    ПроцессорВывода.УстановитьДокумент(Таб);
    ПроцессорВывода.Вывести(ПроцессорКомпоновкиДанных, Истина);

    ИмяФайла = КаталогВременныхФайлов() + ".xls";

    Таб.Записать(ИмяФайла,ТипФайлаТабличногоДокумента.XLSX);
    Таб.Записать("C:\Users\123\Desktop\Обработки\Дебиторка .xls", ТипФайлаТабличногоДокумента.XLS);
```

Значение параметра задал:

```1c
    ТипЦен = Настройки.ПараметрыДанных.Элементы.Найти("ТипЦен");
    ТипЦен.Значение = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("000000354");
```

но почему выводит ошибку:

{ВнешняяОбработка.ВнешняяОбработка1.Форма.Форма.Форма(1245)}: Ошибка при вызове метода контекста (Вывести)
 ПроцессорВывода.Вывести(ПроцессорКомпоновкиДанных, Истина);
по причине:
Ошибка вывода результата
по причине:
Ошибка при выводе результата
по причине:
Ошибка получения данных
по причине:
Ошибка создания набора данных "НаборДанных1"
по причине:
Ошибка при исполнении запроса набора данных
по причине:
{(28, 43)}: Не задано значение параметра "ТипЦен"
ЦеныНоменклатурыСрезПоследних.ТипЦен = <<?>>&ТипЦен) КАК ЗакупочныеЦены
Неправильно задаю значение параметра, если да, то как правильно задать?

Запрос СКД:

```1c
    ВЫБРАТЬ
    РеализацияТоваровУслугТовары.Ссылка.Подразделение,
    РеализацияТоваровУслугТовары.Ссылка.ТорговыйПредставитель,
    РеализацияТоваровУслугТовары.Ссылка.Контрагент,
    РеализацияТоваровУслугТовары.Ссылка.ТорговаяТочка,
    РеализацияТоваровУслугТовары.Номенклатура,
    ЕСТЬNULL(РеализацияТоваровУслугТовары.Количество, 0) КАК Количество,
    ЕСТЬNULL(РеализацияТоваровУслугТовары.Ссылка.СуммаДокумента, 0) КАК Продажа,
    РеализацияТоваровУслугТовары.Цена КАК ЦенаПродажи,
    (ЕСТЬNULL(РеализацияТоваровУслугТовары.Цена, 0) - ЗакупочныеЦены.Цена) / ЗакупочныеЦены.Цена * 100 КАК Наценка,
    ЗакупочныеЦены.Цена
    ИЗ
    Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
    ЛЕВОЕ СОЕДИНЕНИЕ (ВЫБРАТЬ
    ЕСТЬNULL(ЦеныНоменклатурыСрезПоследних.Цена, 0) * 0.83 КАК Цена,
    ЦеныНоменклатурыСрезПоследних.Номенклатура КАК Номенклатура
    ИЗ
    РегистрСведений.ЦеныНоменклатуры.СрезПоследних КАК ЦеныНоменклатурыСрезПоследних
    ГДЕ
    ЦеныНоменклатурыСрезПоследних.ТипЦен = &ТипЦен) КАК ЗакупочныеЦены
    ПО РеализацияТоваровУслугТовары.Номенклатура = ЗакупочныеЦены.Номенклатура
    ГДЕ
    РеализацияТоваровУслугТовары.Ссылка.Дата МЕЖДУ НАЧАЛОПЕРИОДА(&НачалоПериода, ДЕНЬ) И КОНЕЦПЕРИОДА(&КонецПериода, ДЕНЬ)
```
---
видимо, еще признак "Использование" нужно в Истина установить
---
//Настройки = СхемаКомпоновкиДанных.НастройкиПоУмолчанию;
       Настройки = СхемаКомпоновкиДанных.КомпоновщикНастроек.Настройки;
---
{ВнешняяОбработка.ВнешняяОбработка1.Форма.Форма.Форма(995)}: Поле объекта не обнаружено (КомпоновщикНастроек)
 Настройки = СхемаКомпоновкиДанных.КомпоновщикНастроек.Настройки;
---
видимо, еще признак "Использование" нужно в Истина установить
---
Выдает ошибку
Использование не является объектным типом
---
СхемаКомпоновкиДанных.Параметры.ТипЦен.Значение = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("000000354");
---
спасибо
---
вернитесь к своєму варианту, только замените
    ТипЦен = Настройки.ПараметрыДанных.Элементы.Найти("ТипЦен");
    ТипЦен.Значение = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("000000354");
на то что сверху написал.
---
все , правильно, в первых я неправильно сделал)

```1c
    ТипЦен = Настройки.ПараметрыДанных.Элементы.Найти("ТипЦен");
    ТипЦен.Значение = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("000000354");
    ТипЦен.Использование=Истина;
```

Спасибо